
高齢者福祉の財政負荷という指標を設定し

た。 

経済環境スイッチを付加。　ONで30年後に

外部標準賃金と原材料費の増加が止まるよ


Budget

Budjet = Budjet_Need_Basic+Elderies_Welfare_Cost_add

Care_Stf = (Care_Stf_ND_T)*高齢者福祉政策

Local_Tax = Local_Corp_Tax+Local__Incom_Tax_T

財政力指数 = Basic_Tax_Incom/

((Budjet_Need_Basic+DELAY(Budjet_Need_Basic,1)+DELAY(Budjet_Need_Basic,2))/3)

ドキュメント： budjet Index

Day_Serv_Cent = Day_Serv_Cent_ND*高齢者福祉政策

Budjet_balance = Budjet_Need_Basic/Budjet

Local_Corp_Tax = Max(Profit*Locla_Corp_Tax_R, 0)

高齢者福祉政策 = If 高齢者対策_SWT=1 and TIME>=30 then 高齢者福祉政策指標 else 1

ドキュメント： 　Elderies Welfaare PC

Basic_Tax_Incom = Local_Tax*0.75

Budget_Need_Unit = 0.25

Bujdet_Needs_Crt = 1

Locla_Corp_Tax_R = 0.5

Budjet_Need_Basic = Budget_Need_Unit*Bujdet_Needs_Crt*popT

Care_stf_Cost_STD = 3000

ドキュメント： 

平成３（1991）年度　家事型　1,683,562円   身体介護型　2,525,344円

1994年　年額　3,346,502円

ホームヘルパー一人あたりの国の事業関連予算

1988　332,337 円

1990   573,150　円

1995   591364　円

これの（２倍＋利用者負担＋市町村持ちだし分）×ホームヘルパー数で計算できるはずだ。

　しかし、武田、1995、p.70のホームヘルプ事業国庫支出金単価の表から考え、ホームヘルパーの手当＝事業費と考えて良さそうだ。

Day_Serv_Cent_STD = 18670+2074

ドキュメント： 平成３年度　B型　18,670千円／年＋（訪問事業の入浴サービスを実施する場合）2,074千円／年　厚生省、1992、p.110

Local_Incom_Tax_R = 0.1

Respite_Care_Serv = Respite_Care_Serv_ND*高齢者福祉政策

Budjet__f_Elderies = Elderies_Welfaare_Budjet_int*Elderies_Welfare_Cost_mlt

Local__Incom_Tax_T = 就業人口*賃金*Local_Incom_Tax_R

高齢者福祉政策指標 = 2

Local_Alocation_Tax = Max(Budjet_Need_Basic-Basic_Tax_Incom,0)

Service_Center_Cost = Day_Serv_Cent*((Day_Serv_Cent_STD)/Care_stf_Cost_STD)

ドキュメント：  平成３年度　B型　18,670千円／年＋（訪問事業の入浴サービスを実施する場合）2,074千円／年　厚生省、1992、p.110

在宅介護支援センター

ソーシャルワーカー（または介護福祉士）　１名

保健婦（または看護婦）１名

の計２名を必要とする。

年間運営費　9,337,000円／年

年間施設運営費＝サービスセンター数×賃金×５として設定。

これは、仮定の数値なので、資料にあたり要修正。

Elderies_Welfare_Cost = (Care_Stf+Service_Center_Cost+Respite_Care_Serv_Cost)*賃金

Respite_Care_Serv_Cost = Respite_Care_Serv*(5/Care_stf_Cost_STD)

ドキュメント： ショートステイ事業費（特別養護老人ホームなど）

1986年　5000円／人日

1990年　5460円／人日

199４年　6080円／人日

武田　宏、1995、・p.111ー112　ショートステイの事業費

平均所得を350万円と仮定　5000/3500000=0.00142

ショートスティサービスの目標量の設定×5000？

Elderies_Welfare_Cost_add = Elderies_Welfaare_Budjet_int*(Elderies_Welfare_Cost_mlt-1)

Elderies_Welfare_Cost_mlt = Elderies_Welfare_Cost/INIT(Elderies_Welfare_Cost)

Elderies_Welfaare_Budjet_int = INIT(Budjet_Need_Basic)*0.1*0.7*0.3

Elderies_Welfare_PCmlt = グラフ(高齢者福祉政策)

(0.00, 2.00), (0.2, 1.80), (0.4, 1.60), (0.6, 1.40), (0.8, 1.20), (1.00, 1.00), (1.20, 0.8), (1.40, 0.6), (1.60, 0.4), (1.80, 0.2), (2.00, 0.00)

ドキュメント： 高齢者福祉政策倍数

1.生産年齢人口の移動率＜0  生産年齢人口が流出の場合にのみ有効

高齢者福祉政策＝0　全く、何の施策も取らない。＝流出を促進２（生産年齢人口と同じ流出率）

高齢者福祉政策＜１　福祉需要を満たし切らない。＝流入を抑え流出を促進（生産年齢人口の流出率の半分より大きくなる）　1-2

高齢者福祉政策=１　福祉需要を国の基準で満たす。＝影響力なし（生産年齢人口の流出率の半分）1

高齢者福祉政策＞１福祉需要を国の基準以上で満たす。＝流出を抑え流入を促進（生産年齢人口の流出率の半分以下、最大０）　0

Economic Environment

price(t) = price(t - dt) + (price_change) * dt

初期値 price = 賃金

インフロー：

price_change = price*price_change_R

demand(t) = demand(t - dt) + (demand_cahnge) * dt

初期値 demand = 生産量

インフロー：

demand_cahnge = demand*demand_cahnge_R

標準賃金(t) = 標準賃金(t - dt) + (standard_wage_change) * dt

初期値 標準賃金 = price

ドキュメント： standard wage

インフロー：

standard_wage_change = 標準賃金*standard_wage_change_R*経済環境変化

material_price(t) = material_price(t - dt) + (material__price_change) * dt

初期値 material_price = price*0.4

インフロー：

material__price_change = material_price*material_price_change_R*経済環境変化

wage_raio = IF TIME>=30 AND 賃金格差補償SWT=1 then 賃金格差補償率 else 賃金/標準賃金 

ドキュメント： オリジナル：wage/standard_wage

オプション：

経済環境変化 = IF 経済環境変化SWT=1 and TIME>=30 then 0 else 1

必要就業者数 = (demand-stock)/productivity

ドキュメント： disired employee

employee_ratio = 必要就業者数/就業人口

price_change_R = 0.01

wage_change__R = 0.01

賃金格差補償率 = 1

demand_cahnge_R = 0.01

経済環境変化SWT = 1

賃金格差補償SWT = 0

price_woring_indicator = price-material_price

standard_wage_change_R = 0.01

material_price_change_R = 0.01

inmig_m_f_emp = グラフ(employee_ratio)

(0.00, 0.00), (0.5, 0.5), (1.00, 1.00), (1.50, 1.50), (2.00, 2.00), (2.50, 2.50), (3.00, 3.00), (3.50, 3.50), (4.00, 4.00), (4.50, 4.50), (5.00, 5.00)

inmig_m_f_wage = グラフ(wage_raio)

(0.00, 0.00), (0.2, 0.00), (0.4, 0.00), (0.6, 0.00), (0.8, 0.00), (1.00, 1.00), (1.20, 1.00), (1.40, 1.00), (1.60, 1.00), (1.80, 1.00), (2.00, 1.00)

outmig_m_f_emp = グラフ(1/employee_ratio)

(0.00, 0.00), (0.5, 0.5), (1.00, 1.00), (1.50, 1.50), (2.00, 2.00), (2.50, 2.50), (3.00, 3.00), (3.50, 3.50), (4.00, 4.00), (4.50, 4.50), (5.00, 5.00)

outmig_m_f_wage = グラフ(1/wage_raio)

(0.00, 1.00), (0.5, 1.00), (1.00, 1.00), (1.50, 1.50), (2.00, 2.00), (2.50, 2.50), (3.00, 3.00), (3.50, 3.50), (4.00, 4.00), (4.50, 4.50), (5.00, 5.00)

Education

Class = pop_0_to_14/Class_STD

JOb_In = Up_Sec_Schl__3-(Univ_Ad+Job_Out)

Job_Out = (Up_Sec_Schl__3-Univ_Ad)*(1-Job_stay_R)

ドキュメント：  地域外への高卒就職者数＝（高校卒業者-高等教育進学者）＊（１ー地元就職率）

Univ_Ad = Up_Sec_Schl__3*Univ_Ad_R

Class_Int = INIT(pop_0_to_14)/INIT(Class_STD)

Class_STD = 40

Job_stay_R = 0.5+migraton_R_15_to_64

ドキュメント： 地元就職率の標準値を５０％として、これに生産年齢人口の移動率の影響を加える。

生産年齢人口の移動率が正なら（人口流入）、地元就職率は５０％以上となり、

生産年齢人口の移動率が負なら（人口流出）、地元就職率は５０％以下となる。

Class_Ratio = Class/Class_Int

Sec_Schl__Sum = SUM(Up_Sec_Schl__1,Up_Sec_Schl__2,Up_Sec_Schl__3)

Class_Sec_Schl = ((Sec_Schl__Sum/Class_Sec_Schl_STD)/3)

Up_Sec_Schl__1 = matur_14_to_15*(1-migraton_R_0_to_14)*Upper_Secondary_Schl_R*(1-Upper_Secondary_Schl_R_Out)

Up_Sec_Schl__2 = DELAY(Up_Sec_Schl__1,1)*(1-migraton_R_0_to_14-mortal_15_to_64)

Up_Sec_Schl__3 = DELAY(Up_Sec_Schl__2,1)*(1-migraton_R_0_to_14-mortal_15_to_64)

Class_Sec_Schl_Int = INIT(Sec_Schl__Sum)/INIT(Class_Sec_Schl_STD)/3

Class_Sec_Schl_STD = 40

Class_Ratio_Sec_Schl = Class_Sec_Schl/Class_Sec_Schl_Int

Upper_Secondary_Schl_R_Out = 0.1

Univ_Ad_R = グラフ(TIME)

(0.00, 0.156), (5.00, 0.21), (10.0, 0.209), (15.0, 0.288), (20.0, 0.322), (25.0, 0.322), (30.0, 0.322), (35.0, 0.322), (40.0, 0.322), (45.0, 0.322), (50.0, 0.322)

Upper_Secondary_Schl_R = グラフ(TIME)

(0.00, 0.655), (5.00, 0.832), (10.0, 0.918), (15.0, 0.963), (20.0, 0.963), (25.0, 0.963), (30.0, 0.963), (35.0, 0.963), (40.0, 0.963), (45.0, 0.963), (50.0, 0.963)

Employment

Un_emp(t) = Un_emp(t - dt) + (unemplyed_chg) * dt

初期値 Un_emp = (pop_15_to_64*labor_force_RN1564+pop_65_plus*Labor_force_RN65)*unemployed_RN

インフロー：

unemplyed_chg = ((就業人口+Un_emp)*unemployed_RN)-Un_emp

ドキュメント： 1999-8-13　　式を訂正。

＊ここでいうunemployed_RNは、（就業人口＋失業人口）に占める失業人口の割合

就業人口(t) = 就業人口(t - dt) + (emplyed_change - unemplyed_chg) * dt

初期値 就業人口 = pop_15_to_64*labor_force_RN1564+pop_65_plus*Labor_force_RN65-Un_emp

ドキュメント： employed

インフロー：

emplyed_change = pop_change_15_to_64*labor_force_RN1564+pop_change_65plus*Labor_force_RN65

アウトフロー：

unemplyed_chg = ((就業人口+Un_emp)*unemployed_RN)-Un_emp

ドキュメント： 1999-8-13　　式を訂正。

＊ここでいうunemployed_RNは、（就業人口＋失業人口）に占める失業人口の割合

unemp_R = PCT(Un_emp/(就業人口+Un_emp))

Job_In_Ratio = PCT(JOb_In/就業人口)

unemployed_RN = 0.01

labor_force1564 = pop_15_to_64*labor_force_RN1564

labor_force_STR = PCT(labor_force_65plus/(labor_force1564+labor_force_65plus))

Labor_force_RN65 = 0.2

labor_force_65plus = pop_65_plus*Labor_force_RN65

labor_force_RN1564 = 0.6

Industry

賃金(t) = 賃金(t - dt) + (wage_change) * dt

初期値 賃金 = 1

ドキュメント： wage

インフロー：

wage_change = 賃金*current_profit_rate

ドキュメント： オリジナル＝wage*current_profit_rate

オプション＝(Wage_Level-wage)

stock(t) = stock(t - dt) + (生産量 - sales) * dt

初期値 stock = 0

インフロー：

生産量 = 就業人口*productivity

ドキュメント： production

アウトフロー：

sales = demand

Profit(t) = Profit(t - dt) + (revenues - expences) * dt

初期値 Profit = revenues*0.1

インフロー：

revenues = If TIME>=30   then sales*(price+価格補助金*価格補助制度SWT) Else sales*price

ドキュメント： オリジナル：sales*price

オプション：If TIME>=30   then sales*(price+価格補助率*価格補助制度SWT) Else sales*price

アウトフロー：

expences = investment+material_cost+wage_payment

productivity(t) = productivity(t - dt) + (investment_effect - retirement_effect) * dt

初期値 productivity = 賃金*2

インフロー：

investment_effect = productivity*investment_performance+0.01*技術革新速度*技術革新効果

ドキュメント： オリジナル：productivity*investment_performance*technology_mult

オプション1：productivity*investment_performance+0.01*technology_mult

オプション2：if wage_raio>=1 then productivity*investment_performance  else  productivity*investment_performance+0.01*technology_mult

アウトフロー：

retirement_effect = productivity*retirement__R

investment = revenues*investment_RT

価格補助金 = 0

wage_payment = 就業人口*賃金

技術革新効果 = if  TIME>=30 and 技術革新政策SWT=1 then 1 else 0

技術革新速度 = 0

material_cost = material_price*生産量

retirement__R = 0.01

investment_mult = investment/INIT(investment)

価格補助制度SWT = 0

技術革新政策SWT = 0

current_profit_rate = (revenues-expences)/revenues

profit_revenue_ratio = Profit/revenues

stock_worning_indicator = stock/demand

investment_RT = グラフ(profit_revenue_ratio)

(0.00, 0.00), (0.1, 0.1), (0.2, 0.1), (0.3, 0.1), (0.4, 0.1), (0.5, 0.1), (0.6, 0.1), (0.7, 0.1), (0.8, 0.1), (0.9, 0.1), (1, 0.1)

investment_performance = グラフ(investment_mult)

(0.00, 0.00), (1.00, 0.01), (2.00, 0.0242), (3.00, 0.0336), (4.00, 0.0399), (5.00, 0.0438), (6.00, 0.0459), (7.00, 0.0473), (8.00, 0.049), (9.00, 0.0497), (10.0, 0.0504)

Population

pop_0_to_14(t) = pop_0_to_14(t - dt) + (総出生力 + migration_0_to_14 - matur_14_to_15 - deaths__0_to_14) * dt

初期値 pop_0_to_14 = 3150

ドキュメント： オリジナル＝3150

補正値＝3147

インフロー：

総出生力 = pop_15_to_64*出生力

ドキュメント： total Births

migration_0_to_14 = pop_0_to_14*migraton_R_0_to_14

アウトフロー：

matur_14_to_15 = (pop_0_to_14*(1-mortal_0_to_14))/15

ドキュメント： オリジナル＝(pop_0_to_14*(1-mortal_0_to_14))/15

オプション＝pop_0_to_14/15

deaths__0_to_14 = pop_0_to_14*mortal_0_to_14

ドキュメント： オリジナル＝pop_0_to_14*mortal_0_to_14

pop_65_plus(t) = pop_65_plus(t - dt) + (matur_64_to_65 + migration_65plus - deaths_65_plus) * dt

初期値 pop_65_plus = 2500

ドキュメント： オリジナル＝2500

インフロー：

matur_64_to_65 = pop_15_to_64*(1-mortal_15_to_64)/50

ドキュメント： オリジナル＝pop_15_to_64*(1-mortal_15_to_64)/50

オプション＝pop_15_to_64/50

migration_65plus = pop_65_plus*migration_R_65_puls

アウトフロー：

deaths_65_plus = pop_65_plus*mortal_65_plus

ドキュメント： オリジナル＝pop_65_plus*mortal_65_plus

pop_15_to_64(t) = pop_15_to_64(t - dt) + (matur_14_to_15 + migration_15_to_64 - deaths_15_to_64 - matur_64_to_65) * dt

初期値 pop_15_to_64 = 10000

ドキュメント： オリジナル＝10000

インフロー：

matur_14_to_15 = (pop_0_to_14*(1-mortal_0_to_14))/15

ドキュメント： オリジナル＝(pop_0_to_14*(1-mortal_0_to_14))/15

オプション＝pop_0_to_14/15

migration_15_to_64 = pop_15_to_64*migraton_R_15_to_64

アウトフロー：

deaths_15_to_64 = pop_15_to_64*mortal_15_to_64

ドキュメント： オリジナル＝pop_15_to_64*mortal_15_to_64

matur_64_to_65 = pop_15_to_64*(1-mortal_15_to_64)/50

ドキュメント： オリジナル＝pop_15_to_64*(1-mortal_15_to_64)/50

オプション＝pop_15_to_64/50

出生力 = 0.021315*fertility_mult

ドキュメント： fertility

オリジナル＝0.021315*fertility_mult

inmig_mult = inmig_m_f_wage*inmig_m_f_emp*生活環境指標からの移動率倍数*inmig_RN_15_to_64

outmig_mult = outmig_m_f_wage*outmig_m_f_emp*(1/生活環境指標からの移動率倍数)*outmig_RN_15_to_64

ドキュメント： オリジナル：outmig_m_f_wage*outmig_m_f_emp*(1/Liv_Index_Mlt)*outmig_RN_15_to_64

オプション：IF 転出倍数スイッチ=1 then outmig_m_f_wage*outmig_m_f_emp*(1/Liv_Index_Mlt)*outmig_RN_15_to_64 Else 0

mortal_0_to_14 = 0.001

mortal_65_plus = 0.08*mortal_65_plus__mult

高齢者対策_SWT = 0

mortal_15_to_64 = 0.001

inmig_RN_15_to_64 = 0.1

migraton_R_0_to_14 = migraton_R_15_to_64*migraton_R_0_to_14_mult

outmig_RN_15_to_64 = 0.1

老年人口移動率倍数 = 0.5

ドキュメント： migration R65 plus mult

migration_R_65_puls = IF migraton_R_15_to_64>=0 then migraton_R_15_to_64*老年人口移動率倍数 else migraton_R_15_to_64*老年人口移動率倍数*Elderies_Welfare_PCmlt+高齢者福祉政策*0.01*高齢化対策の移動効果

ドキュメント： オリジナル：IF migraton_R_15_to_64>=0 then migraton_R_15_to_64*migration_R_65_plus_mult else migraton_R_15_to_64*migration_R_65_plus_mult*Elderies_Welfare_PCmlt

オプション：IF migraton_R_15_to_64>=0 then migraton_R_15_to_64*migration_R_65_plus_mult else migraton_R_15_to_64*migration_R_65_plus_mult*Elderies_Welfare_PCmlt+高齢者福祉政策*0.01*高齢者対策人口移動効果

migraton_R_15_to_64 = inmig_mult-outmig_mult

高齢化対策の移動効果 = IF 高齢者対策_SWT=1 and TIME>=30 and  高齢者福祉政策 > 2 then 1 else 0

migraton_R_0_to_14_mult = 1

fertility_mult = グラフ(TIME)

(0.00, 1.00), (5.00, 0.9), (10.0, 0.8), (15.0, 0.7), (20.0, 0.6), (25.0, 0.5), (30.0, 0.5), (35.0, 0.5), (40.0, 0.5), (45.0, 0.5), (50.0, 0.5)

mortal_65_plus__mult = グラフ(TIME)

(0.00, 1.00), (5.00, 0.9), (10.0, 0.8), (15.0, 0.7), (20.0, 0.6), (25.0, 0.5), (30.0, 0.5), (35.0, 0.5), (40.0, 0.5), (45.0, 0.5), (50.0, 0.5)

Retail

Retail_Shops(t) = Retail_Shops(t - dt) + (- Retail_Shops_chng) * dt

初期値 Retail_Shops = 100

アウトフロー：

Retail_Shops_chng = 1-Retai__Shops_chng_Mult

Retail_Per_CPT = Retail_Sales_Volueme/popT

Retail_Staff_STD = 2.9

Local_Incom_Net_T = IF 賃金格差補償SWT=1 and TIME>=30  then 就業人口*標準賃金*賃金格差補償率-Local__Incom_Tax_T ELSE 就業人口*賃金-Local__Incom_Tax_T 

ドキュメント： オリジナル：employed*wage-Local__Incom_Tax_T

オプション：

Shopping_R_at_Com = 0.25

Retail_Sales_Volueme = Local_Incom_Net_T*Shopping_R_at_Com

Retail_Sales_Per_Shop = Retail_Sales_Volueme/Retail_Shops

Retail_Profit_per_staff = Retail_Sales_Per_Shop*0.2/Retail_Staff_STD

Retai__Shops_chng_Mult = グラフ(Retail_Profit_per_staff/INIT(Retail_Profit_per_staff))

(0.00, 0.00), (0.2, 0.2), (0.4, 0.4), (0.6, 0.6), (0.8, 0.8), (1.00, 1.00), (1.20, 1.20), (1.40, 1.40), (1.60, 1.60), (1.80, 1.80), (2.00, 2.00)

Social Walfare 

Hospital_R = 0.03

ドキュメント： 65歳以上人口の入院率＝３％　厚真町データ

参考：全国　1.63％ (1995)

Day_Serv_ND = 2*46*0.6*(Elderlies_demented_at_H+Elderlies_bed_fast)+1*50*0.30*Elderlies_frail

ドキュメント： デーサービスの目標量の設定

　在宅寝たきり老人＋在宅痴呆性老人

　：週2回×年間４６週×必要度（60％）

　在宅虚弱老人

　：週1回×年間50週×必要度（30％）

（参考）

要介護老人×週2-6回×年間52週×必要度（？）

在宅虚弱老人×週1-2回×年間52週×必要度（？）

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

Care_Stf_ND_T = Home_Care_adv_Cent_Stf*Day_Serv_Cent_ND+Home_help_serv_stf_ND+Physi_restoration_Stf_ND+Visiting_Health_Serv_Stf_ND+Visiting_Nurse_Serv_Stf__ND

Elderlies_frail = pop_65_plus*Elderies_frail_at_H_Ratio*Elderlies_frail_R

Day_Serv_Cent_ND = (Day_Serv_ND)/(5*52*15)

ドキュメント： デーサービス必要施設数数：単位　施設

　設定：一施設の年間サービス供給量＝5日x52週×15人

=デーサービス年間のべ利用日数÷一施設の年間サービス供給量

サービス必要施設数数：単位　施設

　設定：一施設の年間サービス供給量＝5日x52週×15人

=（デーサービス年間のべ利用日数）÷一施設の年間サービス供給量

Elderlies_frail_R = 0.034

Home_help_serv_ND = 3*46*0.7*(Elderlies_demented_at_H+Elderlies_bed_fast)+1*50*0.35*Elderlies_frail

ドキュメント： ホームヘルプサービスの目標量の設定

　在宅寝たきり老人＋在宅痴呆性老人

　：週３回×年間４６週×必要度（７０％）

　在宅虚弱老人

　：週1回×年間50週×必要度（35％）

（参考）

要介護老人×（週３回ー６回）×52週×必要度（？）

虚弱老人数×（週1回ー2回）×52週×必要度（？）

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

ELderies_in_Bed__R = 0.046

Elderlies_bed_fast = pop_65_plus*Elderies_bed_fast_at_H_Ratio*ELderies_in_Bed__R

Residential_Home_R = 0.05

ドキュメント： 65歳以上人口の特別養護老人ホーム入所率＝5％　厚真町データ

（参考）

65歳以上の人口のうち老人ホーム在所者数　北海道　2.8４％、全国1.62％（1993）

特別養護老人ホーム入所率全国　1.17％（1995）

Physi_restoration_ND = pop_65_plus*(1-Hospital_R-Residential_Home_R)*0.03*0.6*52

ドキュメント： 老人保健サービスの目標量

　機能回復訓練　（６５歳以上人口ー特別擁護老人ホーム収容者-入院者）

　　　　　　　　　×出現率（0.03）×必要度（60％）×年間52回

＊６５以上人口×0.032×週２回×26週×必要度（？）

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

　

Respite_Care_Serv_ND = 6*7*0.4*(Elderlies_demented_at_H+Elderlies_bed_fast)+2*7*0.2*Elderlies_frail

ドキュメント： ショートスティサービスの目標量の設定

　在宅寝たきり老人＋在宅痴呆性老人

　：年６回×７日間×必要度（40％）

　在宅虚弱老人

　：年2回×７日間××必要度（20％）

要介護老人×年６回×必要度（？）

在宅虚弱老人×年1-2回×必要度（？）

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

Home_help_serv_stf_ND = Home_help_serv_ND*2/1200

ドキュメント： ホームヘルパー必要マンパワー数：単位　人

　設定：年間サービス目標量x２時間÷年間常勤換算勤務時間（1200時間）

Home_Care_adv_Cent_Stf = 2

ドキュメント： 在宅介護支援センター

ソーシャルワーカー（または介護福祉士）　１名

保健婦（または看護婦）１名

の計２名を必要とする。

年間運営費　9,337,000円／年

Visiting_Nurse_Serv_ND = (Elderlies_bed_fast+Elderlies_demented_at_H)*12*0.25+Elderlies_frail*6*0.25

ドキュメント： 訪問看護サービスの目標量の設定

　　・在宅寝たきり老人：年12回×必要度（25％）

　　　在宅痴呆性老人：年12回×必要度（25％）

　　　在宅虚弱老人：年6回×必要度（25％）

（参考）

（寝たきり老人＋虚弱老人数）×（週1回ー2回）×52週×必要度（？）

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

　　

　　

Elderlies_demented_at_H = pop_65_plus*Elderlies_demented_at_H_ratio*　Elderlies_demented_at_H_R

Physi_restoration_Stf_ND = (Physi_restoration_ND/15)*7*2/1680

ドキュメント：  

機能回復訓練：（サービスのべ回数÷１回あたり利用人数１５人）×利用時間７時間×平均従事保健婦２名÷保健婦の平均年間就業時間（1680時間）

Elderies_frail_at_H_Ratio = 0.829

Visiting_Health_Serv_Stf_ND = Visiting_Health_Care_Serv_ND*2/1680

ドキュメント： 訪問サービスに必要な要員数

訪問指導：サービスのべ回数×訪問時間2時間÷保健婦の平均年間就業時間（1680時間）

Visiting_Nurse_Serv_Stf__ND =  Visiting_Nurse_Serv_ND/840

ドキュメント： 訪問看護：サービスのべ回数÷看護婦一人あたりの年間訪問回数８４０回

　Elderlies_demented_at_H_R = 0.05

Elderies_bed_fast_at_H_Ratio = 0.343

Visiting_Health_Care_Serv_ND = Elderlies_bed_fast*14+Elderlies_frail*8+Elderlies_demented_at_H*3

ドキュメント： 訪問健康サービスの目標量の設定

　　・訪問指導　在宅寝たきり老人：年12回×必要度（100％）

　　　　　　　　在宅虚弱老人：年6回×必要度（100％）

　　　　　　　　在宅痴呆性老人：年3回×必要度（100％）

　　・訪問口腔・栄養指導（各年１回）

　　　　　　　　在宅寝たきり老人：年2回×必要度（100％）

　　　　　　　　在宅虚弱老人：年2回×必要度（100％）

（参考）

痴呆性老人の訪問指導　痴呆性老人×0.372×年3回52週×必要度（？）

訪問指導　（寝たきり老人＋虚弱老人）×年1回×必要度（？）

訪問口腔指導／栄養指導　同上　

武田　宏、1995、p.46-47 市町村老人保健福祉計画作成指針の目標量

＊メモ　この項目だけ、数が合わない？　　

　

Elderlies_demented_at_H_ratio = 0.15

セクターの中ではない

LA(t) = LA(t - dt) + (LF_chg - LA_chg) * dt

初期値 LA = LT*0.1

インフロー：

LF_chg = 0

アウトフロー：

LA_chg = 0

LF(t) = LF(t - dt) + (- LF_chg) * dt

初期値 LF = LT-(LA+LR)

アウトフロー：

LF_chg = 0

LR(t) = LR(t - dt) + (LA_chg) * dt

初期値 LR = LT*0.01

インフロー：

LA_chg = 0

LT = 150

ドキュメント： 総土地面積　単位　km２

全国過疎指定地域の市町村面積の平均値

過疎市町村の土地利用は平成７年度の平均値で、

総面積　150.4　田　6.5　畑8.1　宅地1.4  林野　119.0 その他　15.3

から想定。

popT = SUM(pop_0_to_14,pop_15_to_64,pop_65_plus)

Pop_D = popT/LT

Liv_DF = (SQRT((((LR*1000000)/Pop_FN)/3.14))*2)/(SQRT((((INIT(LR)*1000000)/INIT(Pop_FN))/3.14))*2)

Pop_FN = popT/Pop_FS

ドキュメント： 総世帯数

Liv_D_65 = (SQRT((((LR*1000000)/pop_65_plus)/3.14))*2)/(SQRT((((INIT(LR)*1000000)/INIT(pop_65_plus))/3.14))*2)

test_emp = (就業人口/init(就業人口))*100

交流距離 = Liv_DF*Liv_D_014*Liv_D_65

ドキュメント： LivDis

総移動数 = migration_65plus+migration_0_to_14+migration_15_to_64

Liv_D_014 = (SQRT((((LR*1000000)/pop_0_to_14)/3.14))*2)/(SQRT((((INIT(LR)*1000000)/INIT(pop_0_to_14))/3.14))*2)

Liv_ShopN = (Retail_Shops/Pop_FN)/(INIT(Retail_Shops)/INIT(Pop_FN))

LR_per__P = (LR/popT)*1000000

ドキュメント： 一人当たりの住宅地面積　単位　m２

（住宅地面積／総人口）＊1000000m２

test_Jobin = (JOb_In/INIT(JOb_In))*100

人口減少率 = PCT((popT-INIT(popT))/INIT(popT))

Liv__ShopSV = Retail_Sales_Per_Shop/INIT(Retail_Sales_Per_Shop)

total_deaths = deaths__0_to_14+deaths_15_to_64+deaths_65_plus

財政規模指標 = Budjet/INIT(Budjet)

ドキュメント： liv Budj B

所得政策費用 = if  賃金格差補償SWT=1 and TIME>=30 then 就業人口*(標準賃金*賃金格差補償率-賃金) else 0 

生活環境指標 = (生活基盤の整備指標*買い物の利便利性)/交流距離

ドキュメント： Liv Index

pop_0_to_14_R = (pop_0_to_14/popT)*100

pop_65_plus_R = (pop_65_plus/popT)*100

pop_15_to_64_R = (pop_15_to_64/popT)*100

財政力指数指標 = 財政力指数/INIT(財政力指数)

ドキュメント： Live Bujindex

財政バランス指標 = Budjet_balance/INIT(Budjet_balance)

ドキュメント： Liv Budj Balance

買い物の利便利性 = Liv_ShopN*Liv__ShopSV

ドキュメント： Liv Shop

pop_change_65plus = matur_64_to_65-deaths_65_plus+migration_65plus 

生活基盤の整備指標 = 財政力指数指標*財政規模指標

ドキュメント： Liv Infra

pop_change_15_to_64 = matur_14_to_15-matur_64_to_65-deaths_15_to_64+migration_15_to_64

高齢者福祉の財政負荷 = Budjet__f_Elderies/Budjet_Need_Basic

ドキュメント： 高齢者福祉関連予算と基準財政需要額の比率

Home_Care_Adv_Cent_STD = 9337

ドキュメント： 在宅介護支援センター

ソーシャルワーカー（または介護福祉士）　１名

保健婦（または看護婦）１名

の計２名を必要とする。

年間運営費　9,337,000円／年

価格補助対策の財政負荷 = IF TIME>= 30 then (sales*価格補助金)/Budjet_Need_Basic ELSE 0

所得格差対策の財政負荷 = 所得政策費用/Budjet_Need_Basic

高齢者福祉の財政規模指標 = Budjet__f_Elderies/INIT(Budjet__f_Elderies)

ドキュメント： liv Budj Eld

Pop_FS = グラフ(TIME)

(0.00, 5.00), (3.00, 4.78), (6.00, 4.57), (9.00, 4.32), (12.0, 4.07), (15.0, 3.88), (18.0, 3.70), (21.0, 3.54), (24.0, 3.35), (27.0, 3.18), (30.0, 3.00)

ドキュメント： わが国の平均世帯人員は、1935　5.02　1940　4.99　1945　4.97　1955　4.97　1960　4.54

だから、初期値は　約５人と見て良い。ただし、その後、1965　4.05　1970 3.69　1975　3.45

1980　3.33　1985　3.23　

生活環境指標からの移動率倍数 = グラフ(生活環境指標)

(0.00, 0.5), (0.2, 0.6), (0.4, 0.7), (0.6, 0.8), (0.8, 0.9), (1.00, 1.00), (1.20, 1.10), (1.40, 1.20), (1.60, 1.30), (1.80, 1.40), (2.00, 1.50)

　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　　

